In[1]:=
```
SetDirectory["~/KappaLib/"];
<< kappaLib-1.1.m
<< helper.m
```

```
KappaLib v1.1

Loading helper.m..
```

- **Do coordinate transformation in Metaclass II**

In[4]:=
```
vars = {x0, x1, x2, x3};
```

$$\text{kappa} = \text{emMatrixToKappa}\left[\begin{pmatrix} a1 & -b1 & 0 & 0 & 0 & 0 \\ b1 & a1 & 0 & 0 & 0 & 0 \\ 0 & 0 & a2 & 0 & 0 & -b2 \\ 0 & 1 & 0 & a1 & b1 & 0 \\ 1 & 0 & 0 & -b1 & a1 & 0 \\ 0 & 0 & b2 & 0 & 0 & a2 \end{pmatrix}\right];$$

- **When Theorem 2.1 holds, we have**

a1 = a2 b1 = b2.

We also assume that a1 = 0, that is, we exclude any axion component.

We also assume b1>0.

In[6]:=
```
sub = {a2 → a1, b2 → b1, a1 → 0};
kappa = kappa //. sub;
```

- **Find coordinate transformation that diagonalises g_+^{-1}**

In[8]:=
$$\text{AA} = \begin{pmatrix} 1 & 0 & 0 & b1 \\ 0 & -b1 & 0 & 0 \\ 0 & 0 & -b1 & 0 \\ b1 & 0 & 0 & 0 \end{pmatrix}; \quad \text{(* = gPlus^(-1) *)}$$

$$\text{BB} = \begin{pmatrix} -1 & 0 & 0 & b1 \\ 0 & -b1 & 0 & 0 \\ 0 & 0 & -b1 & 0 \\ b1 & 0 & 0 & 0 \end{pmatrix}; \quad \text{(* = gMinus^(-1) *)}$$

- **Lalt = matrix that diagonalise AA (=eigenvectors of AA)**

In[10]:=
$$\text{subW} = \left\{\text{ww} \to \sqrt{1 + 4\,b1^2}\right\};$$

$$\text{Lalt} = \begin{pmatrix} 0 & 0 & (1-\text{ww})/(2\,b1) & (1+\text{ww})/(2\,b1) \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 \end{pmatrix};$$

In[12]:=
```
FullSimplify[Transpose[Lalt].AA.Lalt //. subW] // MatrixForm
```

Out[12]//MatrixForm=
$$\begin{pmatrix} -b1 & 0 & 0 & 0 \\ 0 & -b1 & 0 & 0 \\ 0 & 0 & \frac{\left(-1+\sqrt{1+4\,b1^2}\right)\left(-1-4\,b1^2+\sqrt{1+4\,b1^2}\right)}{4\,b1^2} & 0 \\ 0 & 0 & 0 & \frac{\left(1+\sqrt{1+4\,b1^2}\right)\left(1+4\,b1^2+\sqrt{1+4\,b1^2}\right)}{4\,b1^2} \end{pmatrix}$$

Define coordinate transformation
Motivation:

3rd matrix (from left) -- diagonalise g+
2nd matrix -- permutate coordinates so that x0 is time for g+
1st matrix -- diagonalise epsilon and mu matrices in kappa

In[13]:=
$$\text{trans} = \text{FullSimplify}\left[\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1/\text{Sqrt}[2] & 1/\text{Sqrt}[2] \\ 0 & 0 & -1/\text{Sqrt}[2] & 1/\text{Sqrt}[2] \end{pmatrix} . \begin{pmatrix} 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{pmatrix} .\text{Inverse}[\text{Lalt}]\right];$$

```
trans//MatrixForm
```

Out[14]//MatrixForm=
$$\begin{pmatrix} \frac{b1}{ww} & 0 & 0 & \frac{-1+ww}{2\,ww} \\ -\frac{b1}{ww} & 0 & 0 & \frac{1+ww}{2\,ww} \\ 0 & \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} & 0 \\ 0 & -\frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} & 0 \end{pmatrix}$$

In[15]:=
```
(* Check that the transformation is orientation preserving *)
FullSimplify[Det[trans]]
```

Out[15]=
$$\frac{b1}{ww}$$

In[16]:=
```
kappaTrans = emCoordinateChange[kappa, trans];
Simplify[ww / b1 emKappaToMatrix[kappaTrans]] // MatrixForm
```

Out[17]//MatrixForm=
$$\begin{pmatrix} 0 & 0 & 0 & -\frac{ww^2}{b1} & 0 & 0 \\ 0 & 0 & -2 & 0 & -2-ww & 0 \\ 0 & 0 & 0 & 0 & 0 & -ww \\ b1 & 0 & 0 & 0 & 0 & 0 \\ 0 & ww & 0 & 0 & 0 & 0 \\ 0 & 0 & -2+ww & 0 & -2 & 0 \end{pmatrix}$$

Compute ABCD matrices

In[18]:=
```
{Amat, Bmat, Cmat, Dmat} = emKappaToABCD[kappaTrans];
```

In[19]:=
```
epsilon = -FullSimplify[Transpose[Amat]];
epsilon // MatrixForm
mu = FullSimplify[Transpose[Inverse[Bmat]]];
mu // MatrixForm
```

Out[20]//MatrixForm=
$$\begin{pmatrix} -\frac{b1^2}{ww} & 0 & 0 \\ 0 & -b1 & 0 \\ 0 & 0 & -\frac{b1\,(-2+ww)}{ww} \end{pmatrix}$$

Out[22]//MatrixForm=
$$\begin{pmatrix} -\frac{1}{ww} & 0 & 0 \\ 0 & -\frac{ww}{2\,b1+b1\,ww} & 0 \\ 0 & 0 & -\frac{1}{b1} \end{pmatrix}$$

Check that the Fresnel surface still decomposes

In[23]:=
```
frTrans = emKappaToFresnel[kappaTrans, vars];
FullSimplify[frTrans]
```

Out[24]=
$$\frac{1}{ww^2} b1^2 \left(b1\,(x0-x1)\,(x0+x1) - ww\left(x2^2+x3^2\right)\right) \left(b1\,(x0-x1)\,\left((-2+ww)\,x0 + (2+ww)\,x1\right) - ww^2\left(x2^2+x3^2\right)\right)$$

In[25]:=

$$\text{AAtrans} = \begin{pmatrix} \frac{b1^2}{ww} & 0 & 0 & 0 \\ 0 & -\frac{b1^2}{ww} & 0 & 0 \\ 0 & 0 & -b1 & 0 \\ 0 & 0 & 0 & -b1 \end{pmatrix};$$

$$\text{BBtrans} = \begin{pmatrix} \frac{b1^2\,(-2+ww)}{ww^2} & \frac{2\,b1^2}{ww^2} & 0 & 0 \\ \frac{2\,b1^2}{ww^2} & -\frac{b1^2\,(2+ww)}{ww^2} & 0 & 0 \\ 0 & 0 & -b1 & 0 \\ 0 & 0 & 0 & -b1 \end{pmatrix};$$

```
delta = frTrans - ww (vars.AAtrans.vars) (vars.BBtrans.vars);
delta = Flatten[CoefficientList[delta, vars]];
delta = simp[delta]
```

Out[29]= {0}

Check

In[30]:=
```
FullSimplify[AAtrans - trans.AA.Transpose[trans]]
FullSimplify[BBtrans - trans.BB.Transpose[trans]]
```

Out[30]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}

Out[31]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}